Examples of SPSS MANOVA Syntax for Univariate & Multivariate Analyses

1. Univariate Between-subjects Analyses

a) One-way analysis with a categorical independent variable ("one-way analysis of variance")

i) Basic instructions

```
manova maths by group (0,3)/print=cellinfo(means)
 /design.
```

ii) Specifying contrasts and testing effect size and homogeneity of variance

```
manova maths by group(0,3)/contrast(group)=simple(1)
 /print=cellinfo(means) param(estim) signif(efsize) homog(all)
 /design.
```

- *simple(1)* specifies contrasts between Group 0 and Group 1, Group 0 and Group 2 and Group 0 and Group 3.

iii) Another way of testing contrasts

```
manova maths by group(0,3)/contrast(group)=simple(1)
 /print=cellinfo(means)
 /design=group(1) group(2) group(3).
```

- *group(1)*, *group(2)* and *group(3)* in the *design* subcommand stand for the first, second and third *group* contrasts respectively.

b) Two-way analysis with two categorical independent variables ("two-way analysis of variance")

i) Basic instructions

```
manova socstud by group (0,3) sex (1,2)/print=cellinfo(means)
 /design.
```

- When no variables are given on the *design* subcommand, MANOVA tests a full factorial model, using the factors named after *by*. In this example, this means the model tested consists of *group*, *sex* and *group by sex*.

ii) Using a sequential analysis to test factors in different orders

```
manova socstud by group (0,3) sex (1,2)/print=cellinfo(means)
 /method=sequential
 /design=group sex/design=sex group.
```

iii) Specifying contrasts and testing a main effects model

```
manova socstud by group (0,3) sex (1,2)/print=param(estim)
 /contrast(group)=polynomial/contrast(sex)=simple(2)/
 /design=group sex.
```

- *Simple(2)* specifies the second level of *sex* as the reference category. It's coincidental that the code for this level is 2. If the codes for *sex* were 0 and 1, and we wanted category 1 to be the reference category, we would still specify *simple(2).*

iv) Testing simple effects

```
manova socstud by group (0,3) sex (1,2)
 /design
 /design=sex group within sex(1) group within sex(2)
 /design=group sex within group(1) sex within group(2)
  sex within group(3) sex within group(4).
```

- *group within sex(1)* tests the overall effect of *group* within the first level of *sex*. Again, if the first level of *sex* was 0, we would still use *sex(1)* to indicate the first level.

iv) Testing simple effects of contrasts

```
manova socstud by group (0,3) sex (1,2)/print=cellinfo(means)
 contrast(group)=polynomial
 /design=sex group(1) within sex(1) group(2) within sex(1)
  group(3) within sex(1)
  group(1) within sex(2) group(2) within sex(2)
  group(3) within sex(2).
```

- *group(1) within sex(1)* tests the first *group* contrast within the first level of *sex*.

c) <u>One continuous independent variable</u> ("bivariate" or "simple" regression)

i) Basic instructions

```
manova english iq/analysis=english/print=param(estim)
 /design=constant iq.
```

- The *analysis* subcommand specifies the dependent variable.
- When there is no *by*, the model must be specified in the *design* subcommand.

ii) Obtaining residual plots

```
manova english iq/analysis=english/print=param(estim)
 /resid=plot/design=constant iq.
```

d) <u>Several continuous independent variables</u> ("multiple regression")

i) Basic instructions

```
manova socstud age iq selfest/analysis=socstud
 /print=param(estim)/design=constant age iq selfest.
```

ii) Obtaining sequential sums of squares, fitting the independent variables in different orders and obtaining effect sizes

```
manova socstud age iq selfest/analysis=socstud/method=sequential
 /print=param(estim) signif(efsize)
 /design=age iq selfest/design=iq age selfest/design=selfest age iq.
```

- Regardless of whether a *unique* or *sequential* analysis is specified, the regression coefficients always show the effect of each variable fitted last.

ii) Testing an interaction between two independent variables

```
compute iqxse=iq * selfest.

manova socstud age iq selfest iqxse/analysis=socstud/method=sequential
 /print=param(estim)
 /design=constant age iq selfest iqxse.
```

- When both components of an interaction are treated as continuous variables (i.e., not specified after the *by* statement), the interaction term must be computed before the *manova* command and included in the list of continuous variables.

ii) Testing all interactions

```
compute iqxse=iq * selfest.
compute iqxage=iq * age.
compute agexse=age * selfest.
compute ixsxa=iq * selfest * age.

manova socstud age iq selfest iqxse iqxage agexse ixsxa
 /analysis=socstud
 /print=param(estim)
 /design=constant age iq selfest iqxse iqxage agexse ixsxa.
```

e) Both categorical and continuous independent variables ("general linear model")

i) Basic instructions (leaving SPSS to create the coding for both the categorical variables)

```
manova maths age iq by group(0,3) cond (1,3)/
  analysis=maths/contrast(group)=simple(1)
  /print=param(estim)
  /design=age iq cond group.
```

- If a mixture of continuous and categorical (i.e., specified after the *by*) variables is used, all variables must be specified in the *design* subcommand if they are to be included in the model.

ii) Basic instructions (creating your own dummy coding for one of the categorical variables)

```
do if (group eq 1).
compute grp1=1.
else.
compute grp1=0.
end if.

do if (group eq 2).
compute grp2=1.
else.
compute grp2=0.
end if.

do if (group eq 3).
compute grp3=1.
else.
compute grp3=0.
end if.

manova maths age iq grp1 grp2 grp3 by cond (1,3)/
 analysis=maths/print=param(estim)
 /design=age iq cond grp1+grp2+grp3.
```

- Using the *do if* method of creating dummy variables means that cases which are missing on the variable for which the dummy variables are being formed (*group* in this case) will also be missing on the dummy variables.
- Putting + between variables in the *design* subcommand means that the variables will be tested as a group in the analysis of variance table (but does not affect the way the regression coefficients are tested – they are always tested as one degree-of-freedom effects).

iii) "Officially" declaring two of the continuous independent variables as covariates and requesting adjusted means ("analysis of covariance")

```
manova maths age iq selfest by group(0,3) cond (1,3)/
 analysis=maths with age iq
 /print=param(estim)
 /pmeans=vars(maths) tables(group cond group by cond)
 /design=selfest cond group.
```

- *pmeans* asks for the means of the dependent variable by *group*, etc, adjusted for the covariates.

iv) Testing an interaction between a categorical and a continuous independent variable ("testing the homogeneity of the regression slopes")

```
manova maths age by group(0,3)/
 analysis=maths
 /print=param(estim)
 /design=age group age by group.
```

- Testing the interaction between *age* and *group* tests whether the relationship between the covariate (*age* in this case) and the dependent variable (*maths*) is the same for each group.

2. Within-subjects Analyses ('Multivariate' layout of data)

a) One-way analysis

i) Basic instructions (two levels of the within-subjects factor: equivalent to a paired *t*-test)

```
manova time1 time2
 /wsfactor=time(2)/print=cellinfo(means)
 /design.
```

ii) Basic instructions (more than two levels in the within-subjects factor)

```
manova time1 time2 time3
 /wsfactor=time(3)/print=cellinfo(means) signif(multiv univ averf gg hf)
 design.
```

- When there are more than two levels in the within-subject factor, two different approaches to within-subject analyses are possible, the univariate (*averf*) and the multivariate (*multiv*). The univariate approach is based on the assumption that the variance of differences between levels of the within-subject factor is the same for any pair of levels. *gg* and *hf* ask for information about the extent of the departure of the data from this assumption. The multivariate approach does not require this assumption.

iii) Specifying contrasts

```
manova time1 time2 time3
 /wsfactor=time(3)/contrast(time)=helmert
 /rename=cons t1v23 t2v3
 /print=cellinfo(means) signif(multiv uni averf gg hf) transform
 design.
```

- The *rename* subcommand applies labels to the contrasts which help when interpreting the output
- *transform* in the *print* subcommand asks for the coefficients used in transforming the dependent variables to be displayed. As well as creating the specified (or default) contrasts for the dependent variables, the coefficients are such that the variables are orthonormalised. This means that the sum of the squares of the coefficients applied to the variables add to one. The transformed variables are then such that a test of sphericity is a test of the assumption that the variances of the differences between the levels of the within-subject factor are homogeneous.
- The *helmert* contrast compares each level of the factor with the mean of succeeding levels. In this case *time* has three levels, so the two contrasts are {2,-1,-1} and {0,1,-1}. The Helmert contrasts are orthogonal. Manova insists that all contrasts on within-subject factors are orthogonal, and if you specify non-orthogonal contrasts, Manova will orthogonalise them, which is why it's a good idea to request the *transform*ation matrix. Difference contrasts (not used here) are also orthogonal, and are therefore acceptable to Manova. They are the opposite to Helmert contrasts, in that they compare each level of the factor with the mean of preceding levels. For *time* in this example, they would be {-1,1,0} and {-1,-1,2}.

a) Two-way analysis

i) Basic instructions

```
manova t1c1 t1c2 t2c1 t2c2 t3c1 t3c2
 /wsfactor=time(3) cond(2)
 /print=cellinfo(means) signif(multiv uni averf gg hf)
 design.
```

- The order of the factor names in the *wsfactor* subcommand reflects the frequency with which the levels of the factors change in the variable list. In this case the first two variables (*t1c1* and *t1c2*) both have the same level of *time* (*t1*) but different levels of *cond* (*c1* and *c2*), so *time* comes before *cond*.

ii) Specifying contrasts

```
manova t1c1 t1c2 t2c1 t2c2 t3c1 t3c2
 /wsfactor=time(3) cond(2)/contrast(time)=helmert
 /rename=cons t1v23 t2v3 c1v2 tc1xc tc2xc
 /print=signif(multiv uni averf gg hf)
 design.
```

iii) Testing simple effects

```
manova t1c1 t1c2 t2c1 t2c2 t3c1 t3c2
 /wsfactor=time(3) cond(2)/contrast(time)=helmert
 /rename=cons t1v23 t2v3 c1v2 tc1xc tc2xc
 /signif(multiv uni averf gg hf)
 /wsdesign=cond time within cond(1) time within cond(2)
 /design
 /wsdesign=time cond within time(1) cond within time(2)
        cond within time(3)
 /design.
```

3. Mixed Analyses ('Multivariate' layout of data)

a) Two levels in the within-subject factor

i) Basic instructions

```
manova t1 t2 by group (0,1)
 /wsfactor=time(2)
 /print=cellinfo(means)
 /design.
```

ii) Testing simple effects

```
manova t1 t2 by group (0,1)
 /wsfactor=time(2)
 /print=cellinfo(means)
 /wsdesign=time
 /design=mwithin group(1) mwithin group(2)
 /wsdesign=mwithin time(1) mwithin time(2)
 /design=group.
```

- *mwithin* is used because we are testing the effect of a within-subject factor (*time*) within each level of a between-subject factor, *group* (then the effect of a between-subject factor (*group*) within each level of a within-subject factor, *time*). The *m* stands for *mixed*.

b) Three levels in the within-subject factor

i) Basic instructions

```
manova t1 t2 t3 by group (0,1)
 /wsfactor=time(3)
 /print=cellinfo(means) signif(multiv univ averf gg hf)
 /design.
```

ii) Testing simple effects

```
manova t1 t2 t3 by group (0,1)
 /wsfactor=time(3)
 /print=cellinfo(means) signif(multiv univ averf gg hf)
 /wsdesign=time
 /design=mwithin group(1) mwithin group(2)
 /wsdesign=mwithin time(1) mwithin time(2) mwithin time(3)
 /design=group.
```

iii) Testing interaction contrasts

```
manova t1 t2 t3 by group (0,1)
 /wsfactor=time(3)/contrast(time)=helmert
 /wsdesign=time(1) time(2)
 /design=group.
```

- This breaks the interaction down into the *group x time* contrasts. *Group x time(1)* tests whether the difference between the first level of *time* and the mean of the second and third levels of *time* (the first Helmert contrast) is different for *group* 0 and *group* 1. *Group x time(2)* tests whether the second Helmert contrast of time (2^{nd} level versus 3^{rd} level) is different for the two groups.

4. Multivariate Analyses

a) One factor

i) Basic instructions

```
manova test1 test2 test3 by group (1,3)
 /print=cellinfo(means) signif(multiv univ dimenr)
 /discrim=all
 /design.
```

- *discrim=all* asks for information about the discriminant function(s).
- *dimenr* tests the significance of second and later discriminant functions.

ii) Specifying and testing contrasts

```
manova test1 test2 test3 by group (1,3)
 /contrast(group)=simple(1)
 /print=signif(multiv univ dimenr)
 /discrim=all
  design=group(1) group(2).
```

b) Two factors

i) Basic instructions

```
manova test1 test2 test3 by group (1,3) cond (1,2)
 /print=cellinfo(means) signif(multiv univ)
 /discrim=all
 /design.
```

ii) Testing simple effects

```
manova test1 test2 test3 by group (1,3) cond (1,2)/
  print=signif(multiv univ)/
  discrim=all/
  design=cond group within cond(1) group within cond(2)/
  design=group cond within group(1) cond within group(2)
        cond within group(3).
```

c) Including a continuous independent variable

```
manova test1 test2 test3 age by group (1,3) cond (1,2)
 /analysis=test1 to test3
 /print=signif(multiv univ)
 /discrim=all
 /design=age group cond
 /design=age group cond group by cond.
```

d) Multivariate multiple regression

```
manova test1 test2 test3 age iq
 /analysis=test1 to test3
 /print=signif(multiv univ)
 /discrim=all
 /design=age iq.
```

e) Canonical correlation

```
manova test1 test2 test3 with age iq score1
 /print=signif(multiv dimenr eigen) param(estim)/
 /discrim=all
/design.
```

- No *by* is used – variables on both sides of the equation are continuous, so *with* is used rather than *by*.
- *eigen* asks for the canonical correlation coefficients.

f) Profile analysis

```
manova test1 test2 test3 test4 by group (0,3)/
 /transform=repeated
 /rename=cons t1_t2 t2_t3 t3_t4/
 /analysis=(cons/t1_t2 t2_t3 t3_t4)
 /design.
```

- Profile analysis is used with dependent variables which are measured on the same scale but where there is no suggestion of a within-subject factor.
- *transform=repeated* asks for the dependent variables to be transformed into the differences between neighbouring variables. The first transformed variable is the difference between *test1* and *test2*; the second transformed variable is the difference between *test2* and *test3*, and so on.
- *analysis=(cons/t1_t2 t2_t3 t3_t4)* specifies two analyses: a univariate analysis of the average of all dependent variables (*cons*), followed by a multivariate analysis of the differences between the dependent variables. It is the second analysis which tests for differences between the profiles of different groups. A significant multivariate effect means that the pattern of differences between the dependent variables is different over groups.

Alan Taylor
School of Behavioural Sciences
13th June 1997